

NAG Fortran Library Routine Document

G01FMF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01FMF returns the deviate associated with the lower tail probability of the distribution of the Studentized range statistic, via the routine name.

2 Specification

```

real FUNCTION G01FMF(P, V, IR, IFAIL)
INTEGER                IR, IFAIL
real                 P, V

```

3 Description

The externally Studentized range, q , for a sample, x_1, x_2, \dots, x_r , is defined as

$$q = \frac{\max(x_i) - \min(x_i)}{\hat{\sigma}_e}$$

where $\hat{\sigma}_e$ is an independent estimate of the standard error of the x_i . The most common use of this statistic is in the testing of means from a balanced design. In this case for a set of group means, $\bar{T}_1, \bar{T}_2, \dots, \bar{T}_r$, the Studentized range statistic is defined to be the difference between the largest and smallest means, \bar{T}_{largest} and $\bar{T}_{\text{smallest}}$, divided by the square root of the mean-square experimental error, MS_{error} , over the number of observations in each group, n , i.e.,

$$q = \frac{\bar{T}_{\text{largest}} - \bar{T}_{\text{smallest}}}{\sqrt{MS_{\text{error}}/n}}$$

The Studentized range statistic can be used as part of a multiple comparisons procedure such as the Newman-Keuls procedure or Duncan's multiple range test (see Montgomery (1984) and Winer (1970)).

For a Studentized range statistic the probability integral, $P(q; v, r)$, for v degrees of freedom and r groups, can be written as:

$$P(q; v, r) = C \int_0^\infty x^{v-1} e^{-vx^2/2} \left(r \int_{-\infty}^\infty \phi(y) (\Phi(y) - \Phi(y - qx))^{r-1} dy \right) dx$$

where

$$C = \frac{v^{v/2}}{\Gamma(v/2)2^{v/2-1}}, \quad \phi(y) = \frac{1}{\sqrt{2\pi}} e^{-y^2/2} \quad \text{and} \quad \Phi(y) = \int_{-\infty}^y \phi(t) dt.$$

For a given probability p_0 , the deviate q_0 is found as the solution to the equation

$$P(q_0; v, r) = p_0, \tag{1}$$

using C05AZF. Initial estimates are found using the approximation given in Lund and Lund (1983) and a simple search procedure.

4 References

Lund R E and Lund J R (1983) Algorithm AS 190: probabilities and upper quartiles for the studentized range *Appl. Statist.* **32** (2) 204–210

Montgomery D C (1984) *Design and Analysis of Experiments* Wiley

Winer B J (1970) *Statistical Principles in Experimental Design* McGraw-Hill

5 Parameters

- 1: P – *real* *Input*
On entry: the lower tail probability for the Studentized range statistic, p_0 .
Constraint: $0.0 < P < 1.0$.
- 2: V – *real* *Input*
On entry: the number of degrees of freedom, v .
Constraint: $V \geq 1.0$.
- 3: IR – INTEGER *Input*
On entry: the number of groups, r .
Constraint: $IR \geq 2$.
- 4: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL \neq 0 on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If on exit IFAIL = 1, then G01FMF returns 0.0.

IFAIL = 1

On entry, $P \leq 0.0$,
 or $P \geq 1.0$,
 or $V < 1.0$,
 or $IR < 2$.

IFAIL = 2

The routine was unable to find an upper bound for the value of q_0 . This will be caused by p_0 being too close to 1.0.

IFAIL = 3

There is some doubt as to whether full accuracy has been achieved. The returned value should be a reasonable estimate of the true value.

7 Accuracy

The returned solution, q_* , to equation (1) is determined so that at least one of the following criteria apply.

- (a) $|P(q_*; v, r) - p_0| \leq 0.000005$
- (b) $|q_0 - q_*| \leq 0.000005 \times \max(1.0, |q_*|)$.

8 Further Comments

To obtain the factors for Duncan's multiple-range test, equation (1) has to be solved for p_1 , where $p_1 = p_0^{r-1}$, so on input P should be set to p_0^{r-1} .

9 Example

Three values of p , ν and r are read in and the Studentized range deviates or quantiles are computed and printed.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G01FMF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real            P, V, VALQ
      INTEGER          I, IFAIL, IR
*      .. External Functions ..
      real            G01FMF
      EXTERNAL         G01FMF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G01FMF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' P      V      IR      Quantile '
      WRITE (NOUT,*)
      DO 20 I = 1, 3
         READ (NIN,*) P, V, IR
         IFAIL = -1

*
         VALQ = G01FMF(P,V,IR,IFAIL)
*
         IF (IFAIL.EQ.0 .OR. IFAIL.EQ.3) THEN
            WRITE (NOUT,99999) P, V, IR, VALQ
         END IF
      20 CONTINUE
      STOP
*
99999 FORMAT (1X,F5.2,2X,F4.1,1X,I3,1X,F10.4)
      END
```

9.2 Program Data

```
G01FMF Example Program Data
0.95 10.0 5
0.3 60.0 12
0.9 5.0 4
```

9.3 Program Results

G01FMF Example Program Results

P	V	IR	Quantile
0.95	10.0	5	4.6543
0.30	60.0	12	2.8099
0.90	5.0	4	4.2636
